Generating an Inverter Transfer Characteristic in PSPICE

By Brent Nash (bnash@usc.edu) for use in EE 327

Step 1 – Starting PSPICE

1. Find PSPICE on the Desktop or Start Menu. It will most likely be under a folder called "Orcad" on the start menu.

2. Once PSPICE is open, press CTRL-n to get a new netlist file.

Step 2 – Writing the Netlist

*NOTE: See the last page for the complete netlist

1. Do not put anything important on the first line of the file. By default PSPICE comments out the first line. It's best to put a comment there reminding you what this particular PSPICE file simulates.

2. Model your transistors. You will mainly be concerned about N-type and P-type transistors. Most of your transistors of a particular type will share VTo (threshold) and KP (blah) values, but you will need a new .model statement for each transistor that has different VTo and KP values. To model one N-type and one P-type transistor, enter the following two lines in your netlist:

```
.model    NTRAN    NMOS    (VTo = 0.5, KP = 50u)
.model    PTRAN    PMOS    (VTo = -0.5, KP = 20u)
```

The syntax of the .model statement is as follows:

```
.model    Variable-Name    Transistor-Type    (VTo = #, KP = #)
```

3. Lay out your transistors. Transistors are 4-terminal devices. The terminals on MOSFETs are known as the Drain, Gate, Source and Body. For our purposes now, the Body of an N-Type will always be tied to Vss and the Body of a P-Type will always be tied to Vdd.

An inverter consists of two transistors, one N-Type and one P-Type that connect at their Drains. See the diagram below.



Looking at the diagram, we can see that there are 4 nodes in this circuit: Vin, Vout, Vdd, and Vss. For this circuit, Vout will be node 3, Vin will be node 2, Vdd will be node 1, and Vss (by default in PSPICE) is node 0. Transistors also have lengths and widths corresponding to the sizes of their channels that we will need. The above diagram should be laid out in the netlist as follows:

```
M1   3   2   1   1   PTRAN   W=10u   L=2u
M2   3   2   0   0   NTRAN   W=4u    L=2u
```

The syntax for laying out a transistor (as shown above) is as follows:

variable-name drain gate source body transistor-type W=width L=length

4. Labelling leftover nodes. We need to add the following two statements to the netlist:

```
Vdd   1   0   5
Vin   2   0   0
```

Notice that the voltage given to Vin does not matter because we will vary it.

5. Put in the testing parameters. Now, to get an inverter transfer characteristic, we want to output a graph of Vout vs. Vin for our inverter. A DC analysis will allow us to put a voltage on the X-axis of our graph. So to get Vin on the X-axis, we add this line to the netlist:

```
.dc   Vin   0   5   .1
```

The syntax for the DC analysis is as follows:

```
.dc   variable-name   start-value   stop-value   step-size
```

Then, in order to generate the graph and the output file, we must add the following two lines:

```
.op
.probe
```

The .op command writes information to an output file and the .probe command generates a graph for us to use.

Finally, remember to put the following line at the end of the netlist:

```
.end
```

Step 3 – Simulating the Netlist

1. Once you've finished typing your netlist, you need to save it. Go to File -> Save As and save your netlist somewhere where you can find it. MAKE SURE to save it as a .cir file or it will not work.

2. Once the file is saved, go to File -> Close to close the netlist file. Then go to File -> Open and re-open the file you just closed. Because of an error in PSPICE's design, this must be done in order to be able to simulate.

3. When you have the file reopened, go to Simulation -> Run "filename". This should put up a graph on the screen. You'll notice Vin is on the X-axis going from 0 to 5, just like we specified in the .dc statement.

4. Now we must add Vout to the graph. As you may recall, Vout was node 3 in our circuit. Go to Trace -> Add Trace. In the dialog box that pops up, double click on "V(3)" on the left side. This corresponds to the voltage at node 3.

What we've told PSPICE to do is show us the values of Vout on the Y-axis in comparison to the changing value of Vin.

5. To check out specific points on the curve, go to Trace -> Cursor -> Display. Now as you move your mouse along the curve you will be able to see the (x,y) values.

Go to Trace -> Cursor -> Display again to get rid of the cursor.

Complete Netlist

```
*This is my PSPICE inverter

*model the transistors
.model NTRAN NMOS (VTo=0.5, KP=50u)
.model PTRAN PMOS (VTo=-0.5, KP=20u)

*Put the transistors into the circuit
M1 3 2 1 1 PTRAN W=10u L=2u
M2 3 2 0 0 NTRAN W=4u L=2u

*Fill in the remaining nodes
Vdd 1 0 5
Vin 2 0 0

*Do a DC analysis with Vin on the X-Axis
.dc Vin 0 5 .1

*Get an output file and a graph
.op
.probe

.end
```